Sprenkle 1964 Option Pricing Model w/ Num. Greeks [Loxx]
// This source code is subject to the terms of the Mozilla Public License 2.0 at https://mozilla.org/MPL/2.0/
// © loxx

//@version=5
indicator("Sprenkle 1964 Option Pricing Model w/ Num. Greeks [Loxx]", 
     shorttitle ="S1964OPMNG [Loxx]",
     overlay = true, 
     max_lines_count = 500, precision = 4)

if not timeframe.isdaily
    runtime.error("Error: Invald timeframe. Indicator only works on daily timeframe.") 

import loxx/loxxexpandedsourcetypes/4

color darkGreenColor = #1B7E02 
string callString = "Call"
string putString = "Put"

string Continuous = "Continuous"
string PeriodRate = "Period Rate"
string Annual = "Annual"
string SemiAnnual = "Semi-Annual"
string Quaterly = "Quaterly"
string Monthly = "Monthly"

string rogersatch = "Roger-Satchell"
string parkinson = "Parkinson"
string c2c = "Close-to-Close"
string gkvol = "Garman-Klass"
string gkzhvol = "Garman-Klass-Yang-Zhang"
string ewmavolstr = "Exponential Weighted Moving Average"

string timtoolbar= "Time Now = Current time in UNIX format. It is the number of milliseconds that have elapsed since 00:00:00 UTC, 1 January 1970."
string timtoolnow = "Time Bar = The time function returns the UNIX time of the current bar for the specified timeframe and session or NaN if the time point is out of session."
string timetooltrade = "Trading Day = The beginning time of the trading day the current bar belongs to, in UNIX format (the number of milliseconds that have elapsed since 00:00:00 UTC, 1 January 1970)."

ewmavol(float src, int per) => 
    float lambda = (per - 1) / (per + 1)
    float temp = na
    temp := lambda * nz(temp[1], math.pow(src, 2)) + (1.0 - lambda) * math.pow(src, 2)
    out = math.sqrt(temp)
    out
    
rogerssatchel(int per) =>
    float sum = math.sum(math.log(high/ close) * math.log(high / open)
         + math.log(low / close) * math.log(low / open), per) / per
    float out = math.sqrt(sum) 
    out

closetoclose(float src, int per) => 
    float avg = ta.sma(src, per)
    array<float> sarr = array.new<float>(per, 0)
    for i = 0 to per - 1 
        array.set(sarr, i,  math.pow(nz(src[i]) - avg, 2))
    float out = math.sqrt(array.sum(sarr) / (per - 1))
    out 

parkinsonvol(int per)=>
    float volConst = 1.0 / (4.0 * per * math.log(2))
    float sum = volConst * math.sum(math.pow(math.log(high / low), 2), per)
    float out = math.sqrt(sum) 
    out

garmanKlass(int per)=>
    float hllog = math.log(high / low)
    float oplog = math.log(close / open)
    float garmult = (2 * math.log(2) - 1)
    float parkinsonsum = 1 / (2 * per) * math.sum(math.pow(hllog, 2), per)
    float garmansum = garmult / per * math.sum(math.pow(oplog, 2), per)
    float sum = parkinsonsum - garmansum
    float devpercent = math.sqrt(sum)  
    devpercent

gkyzvol(int per)=>
    float gzkylog = math.log(open / nz(close[1]))
    float pklog = math.log(high / low)
    float gklog = math.log(close / open)
    float garmult = (2 * math.log(2) - 1)
    float gkyzsum = 1 / per * math.sum(math.pow(gzkylog, 2), per)
    float parkinsonsum = 1 / (2 * per) * math.sum(math.pow(pklog, 2), per)
    float garmansum = garmult / per * math.sum(math.pow(gklog, 2), per)
    float sum = gkyzsum + parkinsonsum - garmansum
    float devpercent = math.sqrt(sum)  
    devpercent

nd(float x)=> 
    float out = math.exp(-x * x * 0.5) / math.sqrt(2 * math.pi)
    out 

// Boole's Rule
Boole(float StartPoint, float EndPoint, int n)=>
    float[] X = array.new<float>(n + 1 , 0)
    float[] Y = array.new<float>(n + 1 , 0)
    float delta_x = (EndPoint - StartPoint) / n
    for i = 0 to n
        array.set(X, i, StartPoint + i * delta_x)
        array.set(Y, i, nd(array.get(X, i)))
    float sum = 0
    for t = 0 to (n - 1) / 4 
        int ind = 4 * t
        sum += (1 / 45.0) * 
             (14 * array.get(Y, ind) 
             + 64 * array.get(Y, ind + 1) 
             + 24 * array.get(Y, ind + 2) 
             + 64 * array.get(Y, ind + 3) 
             + 14 * array.get(Y, ind + 4)) 
             * delta_x
    sum

// N(0,1) cdf by Boole's Rule
cnd(float x)=> 
    float out = Boole(-10.0, x, 240)
    out

// adaptation from Espen Gaarder Haug
// Sprenkle  formula

convertingToCCRate(float r, float Compoundings)=>
    float ConvertingToCCRate = 0
    if Compoundings == 0 
        ConvertingToCCRate := r
    else
        ConvertingToCCRate := Compoundings * math.log(1 + r / Compoundings)
    ConvertingToCCRate

Sprenkle(string callputflg, float S, float x, float T, float rho,float k, float v)=>
    float d1 = 0
    float d2 = 0
    float Sprenkle = 0

    d1 := (math.log(S / x) + (rho + math.pow(v, 2) / 2) * T) / (v * math.sqrt(T))
    d2 := d1 - v * math.sqrt(T)
  
    if callputflg == callString
        Sprenkle := S * math.exp(rho * T) * cnd(d1) - (1 - k) * x * cnd(d2)
    else
        Sprenkle := (1 - k) * x * cnd(-d2) - S * math.exp(rho * T) * cnd(-d1)

    Sprenkle

eSprenkle(string outputflg, string callputflg, float S, float x, float T, float rho, float k, float v, float dSin)=>
    float dS = 0
    if na(dSin)
        dS := 0.01

    float eSprenkle = 0

    if outputflg == "p" 
        eSprenkle := Sprenkle(callputflg, S, x, T, rho, k, v)

    else if outputflg == "d"  
        eSprenkle := (Sprenkle(callputflg, S + dS, x, T, rho, k, v) 
             - Sprenkle(callputflg, S - dS, x, T, rho, k, v)) / (2 * dS)

    else if outputflg == "e"  
        eSprenkle := (Sprenkle(callputflg, S + dS, x, T, rho, k, v) 
             - Sprenkle(callputflg, S - dS, x, T, rho, k, v)) 
             / (2 * dS) * S / Sprenkle(callputflg, S, x, T, rho, k, v)

    else if outputflg == "g"  
        eSprenkle := (Sprenkle(callputflg, S + dS, x, T, rho, k, v) 
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             + Sprenkle(callputflg, S - dS, x, T, rho, k, v)) / math.pow(dS, 2)

    else if outputflg == "gv"  
        eSprenkle := (Sprenkle(callputflg, S + dS, x, T, rho, k, v + 0.01)
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v + 0.01) 
             + Sprenkle(callputflg, S - dS, x, T, rho, k, v + 0.01) 
             - Sprenkle(callputflg, S + dS, x, T, rho, k, v - 0.01) 
             + 2 * Sprenkle(callputflg, S, x, T, rho, k, v - 0.01) 
             - Sprenkle(callputflg, S - dS, x, T, rho, k, v - 0.01)) / (2 * 0.01 * math.pow(dS ,2)) / 100

    else if outputflg == "gp"  
        eSprenkle := S / 100 * (Sprenkle(callputflg, S + dS, x, T, rho, k, v) 
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             + Sprenkle(callputflg, S - dS, x, T, rho, k, v)) / math.pow(dS, 2)
        
    else if outputflg == "tg"  
        eSprenkle := (Sprenkle(callputflg, S, x, T + 1 / 365, rho, k, v) 
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             + Sprenkle(callputflg, S, x, T - 1 / 365, rho, k, v)) / math.pow(1 / 365, 2)

    else if outputflg == "dddv"  
        eSprenkle := 1 / (4 * dS * 0.01) * (Sprenkle(callputflg, S + dS, x, T, rho, k, v + 0.01) 
             - Sprenkle(callputflg, S + dS, x, T, rho, k, v - 0.01) 
             - Sprenkle(callputflg, S - dS, x, T, rho, k, v + 0.01) + Sprenkle(callputflg, S - dS, x, T, rho, k, v - 0.01)) / 100

    else if outputflg == "v"  
        eSprenkle := (Sprenkle(callputflg, S, x, T, rho, k, v + 0.01) 
             - Sprenkle(callputflg, S, x, T, rho, k, v - 0.01)) / 2

    else if outputflg == "vv" 
        eSprenkle := (Sprenkle(callputflg, S, x, T, rho, k, v + 0.01) 
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             + Sprenkle(callputflg, S, x, T, rho, k, v - 0.01)) / math.pow(0.01, 2) / 10000

    else if outputflg == "vp"  
        eSprenkle := v / 0.1 * (Sprenkle(callputflg, S, x, T, rho, k, v + 0.01) 
             - Sprenkle(callputflg, S, x, T, rho, k, v - 0.01)) / 2

    else if outputflg == "dvdv"  
        eSprenkle := (Sprenkle(callputflg, S, x, T, rho, k, v + 0.01) 
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             + Sprenkle(callputflg, S, x, T, rho, k, v - 0.01))

    else if outputflg == "t"  
        if T <= (1 / 365)
            eSprenkle := Sprenkle(callputflg, S, x, 1E-05, rho, k, v) - Sprenkle(callputflg, S, x, T, rho, k, v)
        else
            eSprenkle := Sprenkle(callputflg, S, x, T - 1 / 365, rho, k, v) - Sprenkle(callputflg, S, x, T, rho, k, v)

    else if outputflg == "r" 
        eSprenkle := (Sprenkle(callputflg, S, x, T, rho + 0.01, k, v)  
             - Sprenkle(callputflg, S, x, T, rho - 0.01, k, v)) / 2

    else if outputflg == "k" 
        eSprenkle := (Sprenkle(callputflg, S, x, T, rho, k + 0.01, v)  
             - Sprenkle(callputflg, S, x, T, rho, k - 0.01, v)) / 2

    else if outputflg == "s"  
        eSprenkle := 1 / math.pow(dS, 3) * (Sprenkle(callputflg, S + 2 * dS, x, T, rho, k, v) 
             - 3 * Sprenkle(callputflg, S + dS, x, T, rho, k, v) 
             + 3 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             - Sprenkle(callputflg, S - dS, x, T, rho, k, v))

    else if outputflg == "dx"  
        eSprenkle := (Sprenkle(callputflg, S, x + dS, T, rho, k, v) 
             - Sprenkle(callputflg, S, x - dS, T, rho, k, v)) / (2 * dS)
    else if outputflg == "dxdx" // Gamma
        eSprenkle := (Sprenkle(callputflg, S, x + dS, T, rho, k, v) 
             - 2 * Sprenkle(callputflg, S, x, T, rho, k, v) 
             + Sprenkle(callputflg, S, x - dS, T, rho, k, v)) / math.pow(dS, 2) 
    eSprenkle

smthtype = input.string("Kaufman", "Heikin-Ashi Better Caculation Type", options = ["AMA", "T3", "Kaufman"], group = "Spot Price Settings")
srcin = input.string("Close", "Spot Price", group= "Spot Price Settings", 
     options = 
     ["Close", "Open", "High", "Low", "Median", "Typical", "Weighted", "Average", "Average Median Body", "Trend Biased", "Trend Biased (Extreme)", 
     "HA Close", "HA Open", "HA High", "HA Low", "HA Median", "HA Typical", "HA Weighted", "HA Average", "HA Average Median Body", "HA Trend Biased", "HA Trend Biased (Extreme)",
     "HAB Close", "HAB Open", "HAB High", "HAB Low", "HAB Median", "HAB Typical", "HAB Weighted", "HAB Average", "HAB Average Median Body", "HAB Trend Biased", "HAB Trend Biased (Extreme)"])

float K = input.float(275, "Strike Price", group = "Basic Settings")   
string OpType = input.string(callString, "Option type", options = [callString, putString], group = "Basic Settings")   
string side = input.string("Long", "Side", options = ["Long", "Short"], group = "Basic Settings")   

float rho = input.float(6., "% Average Growth Rate", group = "Rates Settings") / 100
string rhocmp = input.string(Continuous, "% Average Growth Rate Compounding Type", options = [Continuous, PeriodRate, Annual, SemiAnnual, Quaterly, Monthly], group = "Rates Settings")
float kv = input.float(0., "% Market Risk Aversion Adjustment", group = "Rates Settings") / 100
string kvcmp = input.string(Continuous, "% Market Risk Aversion Adjustment Compounding Type", options = [Continuous, PeriodRate, Annual, SemiAnnual, Quaterly, Monthly], group = "Rates Settings")
float v = input.float(40., "% Volatility", group = "Rates Settings") / 100

int histvolper = input.int(22, "Historical Volatility Period", group = "Historical Volatility Settings", tooltip =  "Not used in calculation. This is here for comparison to implied volatility")
string hvoltype = input.string(c2c, "Historical Volatility Type", options = [c2c, gkvol, gkzhvol, rogersatch, ewmavolstr, parkinson], group = "Historical Volatility Settings")

string timein = input.string("Time Now", title = "Time Now Type", options = ["Time Now", "Time Bar", "Trading Day"], group = "Time Intrevals", tooltip = timtoolnow + "; " + timtoolbar + "; " + timetooltrade)
int daysinyear = input.int(252, title = "Days in Year", minval = 1, maxval = 365, group = "Time Intrevals", tooltip = "Typically 252 or 365")
float hoursinday = input.float(24, title = "Hours Per Day", minval = 1, maxval = 24, group = "Time Intrevals", tooltip = "Typically 6.5, 8, or 24")

int thruMonth = input.int(3, title = "Expiry Month", minval = 1, maxval = 12, group = "Expiry Date/Time")
int thruDay = input.int(31, title = "Expiry Day", minval = 1, maxval = 31, group = "Expiry Date/Time")
int thruYear = input.int(2023, title = "Expiry Year", minval = 1970, group = "Expiry Date/Time")
int mins = input.int(0, title = "Expiry Minute", minval = 0, maxval = 60, group = "Expiry Date/Time")
int hours = input.int(9, title = "Expiry Hour", minval = 0, maxval = 24, group = "Expiry Date/Time")
int secs = input.int(0, title = "Expiry Second", minval = 0, maxval = 60, group = "Expiry Date/Time")

string txtsize = input.string("Auto", title = "Expiry Second", options = ["Small", "Normal", "Tiny", "Auto", "Large"], group = "UI Options")

string outsize = switch txtsize
    "Small"=> size.small
    "Normal"=> size.normal
    "Tiny"=> size.tiny
    "Auto"=> size.auto
    "Large"=> size.large

// seconds per year given inputs above
int spyr = math.round(daysinyear * hoursinday * 60 * 60)

// precision calculation miliseconds in time intreval from time equals now
start = timein == "Time Now" ? timenow : timein == "Time Bar" ? time : time_tradingday
finish = timestamp(thruYear, thruMonth, thruDay, hours, mins, secs) 
temp = (finish - start) 
float T = (finish - start) / spyr / 1000

kfl=input.float(0.666, title="* Kaufman's Adaptive MA (KAMA) Only - Fast End", group = "Moving Average Inputs")
ksl=input.float(0.0645, title="* Kaufman's Adaptive MA (KAMA) Only - Slow End", group = "Moving Average Inputs")
amafl = input.int(2, title="* Adaptive Moving Average (AMA) Only - Fast", group = "Moving Average Inputs")
amasl = input.int(30, title="* Adaptive Moving Average (AMA) Only - Slow", group = "Moving Average Inputs")

haclose = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, close)
haopen = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, open)
hahigh = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, high)
halow = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, low)
hamedian = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, hl2)
hatypical = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, hlc3)
haweighted = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, hlcc4)
haaverage = request.security(ticker.heikinashi(syminfo.tickerid), timeframe.period, ohlc4)

float rhocmpvalue = switch rhocmp
    Continuous=> 0
    PeriodRate=> math.max(1 / T, 1)
    Annual=> 1
    SemiAnnual=> 2
    Quaterly=> 4
    Monthly=> 12
    => 0

float kvcmpvalue = switch kvcmp
    Continuous=> 0
    PeriodRate=> math.max(1 / T, 1)
    Annual=> 1
    SemiAnnual=> 2
    Quaterly=> 4
    Monthly=> 12
    => 0

float spot = switch srcin
	"Close" => loxxexpandedsourcetypes.rclose()
	"Open" => loxxexpandedsourcetypes.ropen()
	"High" => loxxexpandedsourcetypes.rhigh()
	"Low" => loxxexpandedsourcetypes.rlow()
	"Median" => loxxexpandedsourcetypes.rmedian()
	"Typical" => loxxexpandedsourcetypes.rtypical()
	"Weighted" => loxxexpandedsourcetypes.rweighted()
	"Average" => loxxexpandedsourcetypes.raverage()
    "Average Median Body" => loxxexpandedsourcetypes.ravemedbody()
	"Trend Biased" => loxxexpandedsourcetypes.rtrendb()
	"Trend Biased (Extreme)" => loxxexpandedsourcetypes.rtrendbext()
	"HA Close" => loxxexpandedsourcetypes.haclose(haclose)
	"HA Open" => loxxexpandedsourcetypes.haopen(haopen)
	"HA High" => loxxexpandedsourcetypes.hahigh(hahigh)
	"HA Low" => loxxexpandedsourcetypes.halow(halow)
	"HA Median" => loxxexpandedsourcetypes.hamedian(hamedian)
	"HA Typical" => loxxexpandedsourcetypes.hatypical(hatypical)
	"HA Weighted" => loxxexpandedsourcetypes.haweighted(haweighted)
	"HA Average" => loxxexpandedsourcetypes.haaverage(haaverage)
    "HA Average Median Body" => loxxexpandedsourcetypes.haavemedbody(haclose, haopen)
	"HA Trend Biased" => loxxexpandedsourcetypes.hatrendb(haclose, haopen, hahigh, halow)
	"HA Trend Biased (Extreme)" => loxxexpandedsourcetypes.hatrendb(haclose, haopen, hahigh, halow)
	"HAB Close" => loxxexpandedsourcetypes.habclose(smthtype, amafl, amasl, kfl, ksl)
	"HAB Open" => loxxexpandedsourcetypes.habopen(smthtype, amafl, amasl, kfl, ksl)
	"HAB High" => loxxexpandedsourcetypes.habhigh(smthtype, amafl, amasl, kfl, ksl)
	"HAB Low" => loxxexpandedsourcetypes.hablow(smthtype, amafl, amasl, kfl, ksl)
	"HAB Median" => loxxexpandedsourcetypes.habmedian(smthtype, amafl, amasl, kfl, ksl)
    "HAB Typical" => loxxexpandedsourcetypes.habtypical(smthtype, amafl, amasl, kfl, ksl)
	"HAB Weighted" => loxxexpandedsourcetypes.habweighted(smthtype, amafl, amasl, kfl, ksl)
	"HAB Average" => loxxexpandedsourcetypes.habaverage(smthtype, amafl, amasl, kfl, ksl)
    "HAB Average Median Body" => loxxexpandedsourcetypes.habavemedbody(smthtype, amafl, amasl, kfl, ksl)
	"HAB Trend Biased" => loxxexpandedsourcetypes.habtrendb(smthtype, amafl, amasl, kfl, ksl)
	"HAB Trend Biased (Extreme)" => loxxexpandedsourcetypes.habtrendbext(smthtype, amafl, amasl, kfl, ksl)
	=> haclose

float hvolout = switch hvoltype
    parkinson => parkinsonvol(histvolper)
    rogersatch => rogerssatchel(histvolper) 
    c2c => closetoclose(math.log(spot / nz(spot[1])), histvolper) 
    gkvol => garmanKlass(histvolper) 
    gkzhvol => gkyzvol(histvolper) 
    ewmavolstr => ewmavol(math.log(spot / nz(spot[1])), histvolper) 

if barstate.islast

    sideout = side == "Long" ? 1 : -1

    kouta = convertingToCCRate(rho, rhocmpvalue)
    koutb = convertingToCCRate(kv, kvcmpvalue)

    price =              eSprenkle("p",    OpType, spot, K, T, kouta, koutb, v, na) 
    Delta =              eSprenkle("d",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    Elasticity =         eSprenkle("e",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    Gamma =              eSprenkle("g",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    DgammaDvol =         eSprenkle("gv",   OpType, spot, K, T, kouta, koutb, v, na) * sideout
    GammaP	=            eSprenkle("gp",   OpType, spot, K, T, kouta, koutb, v, na) * sideout
    Vega	=            eSprenkle("v",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    DvegaDvol =          eSprenkle("dvdv", OpType, spot, K, T, kouta, koutb, v, na) * sideout
    VegaP =              eSprenkle("vp",   OpType, spot, K, T, kouta, koutb, v, na) * sideout
    Theta =              eSprenkle("t",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    RhoexpectedRoR =     eSprenkle("r",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    RiskAversion =       eSprenkle("k",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    DDeltaDvol =         eSprenkle("dddv", OpType, spot, K, T, kouta, koutb, v, na) * sideout
    Speed =              eSprenkle("s",    OpType, spot, K, T, kouta, koutb, v, na) * sideout
    DeltaX =             eSprenkle("dx",   OpType, spot, K, T, kouta, koutb, v, na) * sideout
    RiskNeutralDensity = eSprenkle("dxdx", OpType, spot, K, T, kouta, koutb, v, na) * sideout

    var testTable = table.new(position = position.middle_right, columns = 1, rows = 32, bgcolor = color.yellow, border_width = 1)

    table.cell(table_id = testTable, column = 0, row = 0,  text = "Sprenkle 1964 Option Pricing Model", bgcolor=color.yellow, text_color = color.black, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 1,  text = "Option Type: " + OpType, bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 2,  text = "Side: " + side , bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 3,  text = "Spot Price: " + str.tostring(spot, format.mintick) , bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 4,  text = "Strike Price: " + str.tostring(K, format.mintick), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 5,  text = "% Average Growth Rate: " + str.tostring(rho * 100, "##.##") + "%\n" + "Compounding Type: " + rhocmp + "\nCC Rate: " + str.tostring(kouta * 100, "##.##") + "%", bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 6,  text = "% Market Risk Averaion Adjustment: " + str.tostring(kv * 100, "##.##") + "%\n" + "Compounding Type: " + kvcmp + "\nCC Carry: " + str.tostring(koutb * 100, "##.##") + "%", bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 7,  text = "% Volatility (annual): " + str.tostring(v * 100, "##.##") + "%", bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 8,  text = "Time Now: " + str.format("{0,date,MMMM dd, yyyy - HH:mm:ss}", timenow), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 9,  text = "Expiry Date: " + str.format("{0,date,MMMM dd, yyyy - HH:mm:ss}", finish), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)

    table.cell(table_id = testTable, column = 0, row = 10,  text = "Calculated Values", bgcolor=color.yellow, text_color = color.black, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 11,  text = "Hist. Volatility Type: " + hvoltype, bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 12,  text = "Hist. Daily Volatility: " + str.tostring(hvolout * 100, "##.##") + "%", bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 13, text = "Hist. Annualized Volatility: " + str.tostring(hvolout * math.sqrt(daysinyear) * 100, "##.##") + "%", bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)

    table.cell(table_id = testTable, column = 0, row = 14, text = OpType + " Option Price", bgcolor=color.yellow, text_color = color.black, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 15, text = "Price: " + str.tostring(price, format.mintick), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)

    table.cell(table_id = testTable, column = 0, row = 16, text = "Numerical Option Sensitivities", bgcolor=color.yellow, text_color = color.black, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 17, text = "Delta Δ: " + str.tostring(Delta, "##.#####"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 18, text = "Elasticity Λ: " + str.tostring(Elasticity, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 19, text = "Gamma Γ: " + str.tostring(Gamma, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 20, text = "DGammaDvol: " + str.tostring(DgammaDvol, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 21, text = "GammaP Γ: " + str.tostring(GammaP, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 22, text = "Vega: " + str.tostring(Vega, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 23, text = "DVegaDvol: " + str.tostring(DvegaDvol, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 24, text = "VegaP: " + str.tostring(VegaP, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 25, text = "Theta Θ (1 day): " + str.tostring(Theta, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 26, text = "DDeltaDvol: " + str.tostring(DDeltaDvol, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 27, text = "Rho Expected Rate of Return: " + str.tostring(RhoexpectedRoR, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 28, text = "Risk Aversion: " + str.tostring(RiskAversion, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 29, text = "Speed: " + str.tostring(Speed, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 30, text = "Strike Delta: " + str.tostring(DeltaX, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)
    table.cell(table_id = testTable, column = 0, row = 31, text = "Risk Neutral Density: " + str.tostring(RiskNeutralDensity, "##.########"), bgcolor=darkGreenColor, text_color = color.white, text_size = outsize, text_halign = text.align_left)